

Exercise 14

Properties Form

group selection by prop or mat

Groups by PID/MID

Enter the Property name

Prop_1

Enter the Material name

steel

Enter the Group name

fem_elem_select_mat


Create Groups By

> Property ◇ Material ◆ Both

Auto Execute

Select Elements

-Apply- Cancel



Objectives:

- To customize MSC.Patran forms by changing form headings, titles and widgets.



Model Description:

In this exercise, you will edit and customize a PCL program which creates an MSC.Patran form. In particular, you will change the size of the frame, the form title, the headings in the forms and the functionality of the widgets on the form. The file containing the PCL program is named **PROPERTIES_FORM.PCL** and should reside in your working directory.

Files:

All the files that are used in this exercise are listed below. Each list includes the file, where it originated, and a summary of information of how it relates to the exercise.

File	Supplied/Created	Description
properties_form.pcl	Supplied	A group selection form that you will customize.
properties_form.cpp	Created	Created from the C Pre-Processor when the PCL file is run.

Exercise Procedure:

The following steps show how to compile and run the PCL program creating the form.

1. To see what the form looks like, type the following command at the unix prompt:

```
%cpp -P -I/patran/patran3/customization
  properties_form.pcl properties_form.cpp
```

Note: Reference Exercise 12 for the machine specific command for the C Pre-Processor

2. Next, run MSC.Patran by typing patran at the unix prompt or dos command window.
3. Open a New Database called **PCL_cust.db**.

File/New ...


PCL_cust.db

Apply


- In the command window, type the following commands:

```
!!input properties_form.cpp
```

```
ui_exec_function ("properties_form","display")
```



The form should now appear on the display screen.



- Quit MSC.Patran and use a text editor to make the changes to the form as described above. The file you will be changing is properties_form.pcl

Sample Solution:

```

/*$$ Use of PCL in creating customized forms/widgets.
 *
 * Purpose:
 * Show an example of creating a customized form in P3
 *
 * Input:
 * <None>
 *
 * Output:
 * <None>
 *
 * Log:
 *
 * Notes:
 */

#include "appforms.p"

CLASS properties_form

/*
 *Variable initialization
 */

CLASSWIDE widget form_id, @
    dbx1_id, @
    dbx2_id, @
    dbx3_id, @
    apply_button, @
    cancel_button, @
    sort_sw , @
    select_frame, @
    data_box

FUNCTION INIT()

REAL          y_loc, @
              select_data_box_width

WIDGET wid_id

/*
 * Create the form
 */

form_id=ui_form_create( @
/* callback */      "", @
/* x */             FORM_X_LOC, @
/* y */             FORM_Y_LOC, @
/* position */     "UL", @
/* width */        FORM_WID_SML, @
/* height */       FORM_HGT_FULL, @
/* label */        "group selection by prop or mat", @
/* iconname */     "" )

y_loc = FORM_T_MARGIN

```

```

/*
 * Create a descriptive label
 */

ui_label_create(
/* parent */ form_id, @
/* lab_name */ "", @
/* x */ UNFRAMED_L_MARGIN, @
/* y */ y_loc, @
/* lab_text */ "Groups by PID/MID" )

y_loc += LABEL_HGT + INTER_WIDGET_SPACE

/*
 * Create the "Create the Property data box"
 */

dbx1_id= ui_data_box_create(
/* parent */ form_id, @
/* callback */ "", @
/* x */ UNFRAMED_L_MARGIN, @
/* y */ y_loc, @
/* label_length */ 0.0, @
/* box_length */ DBOX_WID_SINGLE, @
/* label */ "Enter the Property name", @
/* value */ "Prop_1", @
/* label_above */ TRUE, @
/* datatype */ "STRING", @
/* num_vals */ 1 )

y_loc += DBOX_HGT_LABOVE+INTER_WIDGET_SPACE

/*
 * Create the "Create the Material data box"
 */

dbx2_id= ui_databox_create(
/* parent */ form_id, @
/* callback */ "", @
/* x */ UNFRAMED_L_MARGIN, @
/* y */ y_loc, @
/* label_length */ 0.0, @
/* box_length */ DBOX_WID_SINGLE, @
/* label */ "Enter the Material name", @
/* value */ "steel ", @
/* label_above */ TRUE, @
/* datatype */ "STRING", @
/* num_vals */ 1 )

y_loc += DBOX_HGT_LABOVE+INTER_WIDGET_SPACE

/*
 * Create the "enter the group name to contain
 * the element data "
 */

dbx3_id= ui_databox_create(
/* parent */ form_id, @
/* callback */ "", @
/* x */ UNFRAMED_L_MARGIN, @
/* y */ y_loc, @
/* label_length */ 0.0, @
/* box_length */ DBOX_WID_SINGLE, @

```

```

/* label */ "Enter the Group name", @
/* value */ "fem_elem_select_mat", @
/* label_above */ TRUE, @
/* datatype */ "STRING", @
/* num_vals */ 1 )

y_loc += DBOX_HGT_LABOV+ INTER_WIDGET_SPACE

/*
* Sorting type switch
*/
sort_sw = ui_switch_create( @
/* parent */ form_id, @
/* callback */ "", @
/* x */ UNFRAMED_L_MARGIN, @
/* y */ y_loc, @
/* num_columns */ 3, @
/* label */ "Create Groups By", @
/* always_one */ TRUE )

ui_item_create( sort_sw, "1", "Property", FALSE )
ui_item_create( sort_sw, "2", "Material", FALSE )
wid_id= ui_item_create( sort_sw, "3", "Both", FALSE )

ui_wid_set( wid_id, "VALUE", TRUE )

y_loc += SWITCH_1R_HGT_LABEL+INTER_WIDGET_SPACE

/*
* select data frame for element selection
*/
select_frame = ui_selectframe_create ( @
/* parent */ form_id, @
/* callback */ "", @
/* Left_margin */ FORM_L_MARGIN, @
/* y */ y_loc, @
/* col_width */ SFRAME_WID_SINGLE, @
/* height */ SFRAME_1SDB_HGT_LABOVE, @
/* always_one */ "Auto Execute" , @
/* recycle */ TRUE )

select_data_box_width=SDBOX_WID_SINGLE-SFRAME_R_MARGIN

/*
* Create the data boxes to be placed inside the dataframe
*/
data_box = ui_selectdatabox_create ( @
/*parent_frame */ select_frame, @
/* callback */ " ", @
/* Left_margin */ SFRAME_L_MARGIN, @
/* y */ SDBOX_Y_LOC1_LABOVE, @
/*label_length */ 0.0, @
/* Box_length */ select_data_box_width, @
/* label */ " Select Elements ", @
/*default_value */ " " , @
/* label_above */ TRUE, @
/* data_type */ "ELEMENT", @
/* prompt */ "select element filter" )

y_loc += SFRAME_LABEL_HGT + SFRAME_1SDB_HGT_LABOVE +@ SFRAME_2EDGE +
INTER_WIDGET_SPACE

```

```

/*
 * Create the "Apply" button
 */
apply_button = ui_button_create(
    /* parent */ form_id,
    /* callback */ "apply_cb",
    /* x */ BUTTON_HALF_X_LOC1,
    /* y */ y_loc,
    /* width */ BUTTON_WID_HALF,
    /* height */ 0.0,
    /* label */ "Apply",
    /* labelinside */ TRUE,
    /* highlight */ TRUE
)

/*
 * Create the "Cancel" button
 */

cancel_button = ui_button_create(
    /* parent */ form_id,
    /* callback */ "cancel_cb",
    /* x */ BUTTON_HALF_X_LOC2,
    /* y */ y_loc,
    /* width */ BUTTON_WID_HALF,
    /* height */ 0.0,
    /* label */ "Cancel",
    /* labelinside */ TRUE,
    /* highlight */ FALSE
)

y_loc += BUTTON_DEFAULT_HGT + FORM_B_MARGIN

ui_wid_set( form_id, "HEIGHT", y_loc )

END FUNCTION

FUNCTION DISPLAY()

    UI_FORM_DISPLAY ( "properties_form" )

END FUNCTION

FUNCTION apply_cb()
    STRING item [20]
    UI_WID_GET (dbx1_id,"VALUE", item)
    ! the value of the data box= `item`
END FUNCTION

FUNCTION cancel_cb()

    UI_FORM_HIDE( "properties_form" )
    select_focus.exit()

END FUNCTION

END CLASS

```

6. To change the title of the form, go to the line that reads:

```
/* label */ "group selection by prop or mat",
```

and change the text to "Group Selection by Prop or Mat" inside the quotation marks.

To change the title of the property heading box, go to the line that reads:

```
/* label */ "Enter the Property name",
```

and change the text to "Enter Property Name" inside the quotation marks. The next line is the default value name given to the property. Change it from "Prop_1" to "Prop1".

Repeat the procedure used to change the property title and default value for the material title ("Enter Material Name"), the default material value ("my_shell") and the group title.

7. Remove the "Both" switch and set the Material switch on by default. To do that, go to the line that reads:

```
wid_id= ui_item_create( sort_sw, "3", "Both", FALSE )
```

and comment the line out as follows:

```
/*wid_id= ui_item_create( sort_sw, "3", "Both", FALSE )*/
```

This blocks the line out and deletes the "Both" switch from the form.

Go to the line above that reads:

```
ui_item_create( sort_sw, "2", "Material", FALSE )
```

and edit the line as follows:

```
wid_id=ui_item_create( sort_sw, "2", "Material", FALSE )
```

This will set the material switch on by default.

-
8. Change the form so that when the property switch is on, the material data box is ghosted out, and when the material switch is on, the property data box is ghosted out. To do this, you have to enter a callback function in the section where the switch is created. Where it reads:

```
/* callback */ "", @
```

edit the line so it reads:

```
/* callback */ "switch_cb", @
```

9. Now, after the command lines

```
UI_FORM_DISPLAY ( "properties_form" )
```

```
END FUNCTION
```

add the following command lines to the file:

```
FUNCTION switch_cb(item1, item2)
  STRING item1 [], item2 []
  IF (item1== "1" && item2 == "ON" ) THEN

    ui_wid_set (dbx2_id, "ENABLE", FALSE)
    ui_wid_set (dbx1_id, "ENABLE", TRUE )

  ELSE IF (item1== "2" && item2 == "ON" ) THEN

    ui_wid_set (dbx1_id, "ENABLE", FALSE)
    ui_wid_set (dbx2_id, "ENABLE", TRUE )

  /*
  * ELSE IF (item1== "3" && item2 == "ON" ) THEN
  *     ui_wid_set (dbx2_id, "ENABLE", TRUE )
  *     ui_wid_set (dbx1_id, "ENABLE", TRUE )
  */
  END IF

END FUNCTION
```

Now, at the UNIX shell or NT command window (see below) type,

```
cpp -IC:\msc\patran90\customization
properties_form.pcl properties_form.cpp
```

10. Next, run p3, and open the database **PCL_cust.db**.

File/Open ...

PCL_cust.db

Apply

Now, type the following commands in the command window:

```
!!input properties_form.cpp
ui_exec_function ("properties_form","display")
```

Your customized form should now look like the following:

group selection by prop or mat

Groups by PID/MID

Enter Property Name

Prop1

Enter Material Name

my_shell

Enter Group Name

fem_elem_select_mat

Create Groups By

Property Material

Auto Execute

Select Elements

-Apply- Cancel

